

Les chaines de caractères


Bonjour à tous ! Aujourd'hui, nous nous retrouvons pour vous apprendre à utiliser les chaînes de caractères en Basic Casio. Celles-ci n'ont, à ma connaissance, pas d'intérêt scolaire, mais elles trouvent une utilité remarquable dans la création de jeux, puisqu'elles sont en effet bien plus rapides que les fonctions traditionnelles du Basic.

Mais que sont-elles vraiment ?

Introduction :

Les chaînes de caractères, *ou strings en anglais*, sont des listes de lettres, de chiffres, de caractères spéciaux. Par exemple, "BONJOUR", "123", "DS42&)-G" sont des chaînes de caractères. Elles sont délimitées par les guillemets anglais « " », que vous ajoutez au programme avec [ALPHA] + [x10].

Attention : les chaînes de caractères ne sont disponibles que sur les calculatrices "récentes", qui apparaissent sur la liste ci-contre. Pour savoir rapidement si vous pouvez les utiliser, faites [SHIFT] / [VARS] (Prgm) / [F6] → [F6]. Si vous avez un onglet « Str », c'est que c'est bon.



```
J'AI LES CHAINES DE
CARACTERES SUR MA CAL
CULATRICE !
```

Liste des calculatrices compatibles :

- Graph 35+ **USB**
- Graph 75
- Graph 85 avec mise à jour 2.xx
- Graph 95 (SD)
- Fx-CG 10/20 (Prizm)

Elles fonctionnent un peu comme des variables classiques, puisque l'on les crée dans un programme selon la syntaxe suivante :

```
"HELLO" → Str 1
"Niveau 42" → Str 2
"Bonjour Darkyz !" → Str 3
```



Attention !

N'oubliez pas les guillemets, sinon la calculatrice affichera une erreur !

La variable spéciale « Str » est disponible dans [VARS] / [F6] / [F5] (Str). Il faut mettre à la suite un nombre entier compris entre 1 et 20. Vous pouvez en effet avoir accès à 20 chaînes de caractères dans un programme.

Les chaînes de caractères peuvent contenir jusqu'à 255 caractères : au delà, la calculatrice vous indique une erreur mémoire. Les espaces et les symboles spéciaux (accessibles avec [F6]) sont aussi comptabilisés dans le nombre de caractères de la chaîne.

Créer une chaîne en fonction de l'utilisateur:

Créer une chaîne, c'est facile : nous l'avons vu, il suffit de mettre des guillemets, une flèche, le mot-clé « Str », et un nombre. Mais que faire si je veux demander à l'utilisateur son nom ?

Et bien c'est encore très simple. Vous vous souvenez du « ? » qui sert à demander un nombre ? Et bien il va nous servir une seconde fois. Du coup, nous pouvons utiliser la syntaxe suivante :

```
"Bonjour !"  
"Quel est ton nom " ? → Str 1
```

A l'intérieur de la chaîne, on aura bien ce qu'a écrit l'utilisateur. Vous pouvez répéter l'opération avec chacune des 20 chaînes, en lui demandant par exemple où il habite, etc.

NB : Si l'utilisateur rentre un nombre, vous ne pourrez pas faire de calculs dessus, puisque les chiffres seront considérés comme des caractères tels que des lettres.

Afficher une chaîne à l'écran:

Pour afficher le contenu d'une chaîne, il existe seulement deux méthodes : utiliser Locate, ou utiliser Text. Dans cette partie, nous ne parlerons pas de la fonction « Text », qui sera vue dans un prochain cours. Toutefois, sachez qu'elle fonctionne de la même manière que Locate.

Revenons-en au Locate. Sa syntaxe, vous l'avez vue précédemment, est :

```
Locate X, Y, Variable (A, B, etc.)
```

Nous pouvons donc l'utiliser avec les chaînes comme ceci, à condition qu'il y ait du texte dans la chaîne :

```
"HELLO WORLD" → Str 1  
Locate 6, 4, Str 1
```

Toutefois, si la chaîne fait plus de 21 caractères de long, elle ne retournera pas à la ligne, et on ne verra donc pas les derniers caractères. Nous allons bientôt voir une astuce pour contourner ce problème.

Opérations et transformations sur les chaînes :

Les chaînes de caractères sont très puissantes dans la mesure où le Basic met à notre disposition 14 fonctions très rapides comparées aux autres fonctions. Vous y accédez via le menu Programme : [SHIFT] / [VAR] / [F6] / [F6] / [F1] (STR).

StrJoin [F1] :

La première commande, StrJoin, sert à relier deux chaînes entre elles. Sa syntaxe est « StrJoin(Str *, Str *) → Str * » avec « * » le numéro des différentes chaînes à utiliser. Vous pouvez aussi intégrer une chaîne entre guillemets comme ceci, par exemple pour dire « Bonjour + Nom du joueur » :

```
"Quel est ton nom " ? → Str 1 // On considère que l'on rentre « Toto »  
StrJoin("Bonjour", Str 1) → Str 1 // Str 1 contient « Bonjour Toto »  
StrJoin(Str 1, " ! ") → Str 1 // Str 1 contient « Bonjour Toto ! »  
ClrText // Pour éviter les traces du texte de la question à l'écran  
Locate 1, 1, Str 1
```

Toutefois, cette fonction est peu utilisée puisqu'elle peut être remplacée par un « + ». Par exemple, la syntaxe suivante est correcte : « Str 1 + Str 2 → Str 3 ». En reprenant le code ci-dessus, on obtient quelque chose de bien plus simple à lire et à comprendre :

```
"Quel est ton nom " ? → Str 1 // On considère que l'on rentre « Toto »  
"Bonjour " + Str 1 + " ! " → Str 1 // Str 1 contient « Bonjour Toto ! »  
ClrText  
Locate 1, 1, Str 1
```



Attention !

N'oubliez pas l'espace entre le « r » de « Bonjour » et le guillemet qui ferme, sinon le texte affiché sera « BonjourToto ! ». Idem avec le « ! ».

StrLen [F2] :

Cette fonction permet de retourner le nombre de caractères dans une chaîne.

```
StrLen("Toto") → A // On peut l'utiliser directement. Ici, A = 4  
StrLen(Str 1) → A // Ou sur des Str.  
StrLen("Bonjour " + Str 1) → A // Et même en mélangeant les deux.
```

StrCmp [F3] :

Cette fonction compare 2 chaînes de caractères (comparaison basée sur le code des caractères). Elle retourne 0 si "chaîne 1" = "chaîne 2", 1 si "chaîne 1" > "chaîne 2" et -1 si "chaîne 1" < "chaîne 2".

Elle s'utilise comme ceci « StrCmp(Str *, Str *) ». Exemples :

```
StrCmp("Toto", "Bonjour") → A // On peut l'utiliser directement
StrCmp(Str 1, Str 2) → A // Ou sur des Str.
If(StrCmp("Bonjour", Str 1)) // Et même en mélangeant les deux et des
conditions.
```

StrSrc [F4] :

Cette fonction cherche la position d'une chaîne dans une autre. La première occurrence est celle qui est retournée. Si la chaîne à chercher n'existe pas dans celle où l'on cherche, la fonction retourne 0.

Sa syntaxe est « StrSrc(Str *, Str *) ». Exemples :

```
StrSrc("ABCD BCDA CDAB", "C") → A // A = 3
StrSrc("Hello Toto!", "Toto") → A // A = 7
StrSrc("Hello Toto!", "A") → A // A = 0
StrSrc(Str 1, Str 2) → A // On peut aussi mettre des Str
```

StrLeft [F6] / [F1] :

Cette fonction récupère le nombre de caractères à partir de la gauche d'une chaîne. Elle s'utilise comme ceci : « StrLeft(Str *, Nombre) → Str * ». Quelques exemples :

```
StrLeft("Bonjour Toto!", 5) → Str 1 // Str 1 = "Bonjo"
StrLeft("Bonjour Toto!", A) → Str 1 // on peut aussi utiliser des variables
StrLeft(Str 1, A) → Str 2 // Et des deux en même temps
```

StrRight [F6] / [F2]:

Idem que StrLeft, mais à partir de la droite. Nous ne s'y attarderons pas, sa syntaxe et son fonctionnement est le même.

StrMid [F6] / [F3]:

Un peu comme StrLeft et StrRight, sauf qu'elle récupère les caractères au milieu d'une chaîne. Sa syntaxe est « StrMid(Str *, A, B) → Str * » avec A la position du premier caractère à récupérer et B le nombre de caractères. Exemples :

```
StrMid("Bonjour Toto!", 5, 6) → Str 1 // Str 1 = "ourTo"
StrMid("Bonjour Toto!", A, 2) → Str 1 // on peut aussi utiliser des
variables
StrMid(Str 1, A, B) → Str 2 // Et des deux en même temps
```

Exp [F6] / [F5]:

Permet de « résoudre » la chaîne : « Exp("12+3") » retourne 15. On peut aussi utiliser plusieurs chaînes : « Exp("12+3" + "-10") » retourne 5.

StrUpr [F6] / [F6] / [F1]:

Met en majuscules la chaîne. « StrUpr("Bonjour Toto") → Str * » renvoie "BONJOUR TOTO".

StrUpr [F6] / [F6] / [F2]:

Met en minuscules la chaîne.

StrInv [F6] / [F6] / [F3]:

Inverse les caractères d'une chaîne. La syntaxe est « StrInv(Str *) → Str * ». Quelques exemples.

```
StrInv("Hello World") → Str 1 // Str 1 = "dlroW olleH"
StrInv(Str 1) → Str 2 // Fonctionne aussi sur les Str *
```

StrShift [F6] / [F6] / [F4]:

Sa syntaxe est « StrShift(Str *, A) → Str * ». Elle supprime le nombre de caractères spécifiés à partir de la gauche. Exemple :

```
StrShift("Bonjour Toto!", 3) → Str 1 // Str 1 = "jourToto!"
```

StrRotate [F6] / [F6] / [F5]:

Cette fonction fait tourner la chaîne sur le nombre de caractères spécifiés. Si ce nombre est positif, la chaîne tourne de droite à gauche, si il est négatif, de gauche à droite. « StrRotate(Str *, A) → Str * ». Exemples :

```
StrRotate("Bonjour Toto!", 3) → Str 1 // Str 1 = "jourToto!Bon"
StrRotate("Bonjour Toto!", -5) → Str 1 // Str 1 = "oto!Bonjour T"
```

On peut aussi, avec cette fonction, faire défiler un texte à l'écran :

```
"Bienvenue sur mon nouveau jeu ! Que voulez vous faire ?" → Str 1
ClrText
While 1=1
StrRotate(Str 1, 1) → Str 1
Locate 1, 4, Str 1
WhileEnd
```

Afficher une chaîne sans limitation de taille :

Maintenant que l'on a vu toutes les fonctions, nous allons pouvoir réaliser le programme qui affichera une chaîne avec des retours à la ligne automatiques. Celui-ci devra afficher le contenu de la Str 1 à l'écran, quelque soit la taille de la chaîne. Ce programme nous sera utile pour afficher un dialogue ou des instructions, par exemple.

Nous aurons besoin de connaître la taille de la chaîne, donc nous utiliserons la fonction « StrLen ». Il nous faudra aussi une boucle qui parcourt la chaîne à chaque ligne. On utilisera donc une boucle « For / To ». De plus, pour afficher ligne par ligne la chaîne, il sera nécessaire de la couper en plusieurs morceaux. La fonction « StrMid » nous sera donc utile. Et pour finir, l'affichage sera réalisé grâce au « Locate ».

Voyons maintenant le code :

```
For 1→A To StrLen(Str 1) Step 21 // On parcourt par ligne de 21 caractères
Locate 1, Int(A÷21)+1, StrMid(Str 1, A , 21) // Et on affiche les 21 caractères
de la ligne en cours.
Next
```

Ce programme vous sera utile, j'en suis sûr !

Conclusion :

Ce chapitre sur les chaînes de caractères est maintenant fini, et nous espérons que vous avez compris l'essentiel des notions abordées. Si ce n'est pas le cas, nous vous conseillons de le relire à tête reposée et avec votre Graph favorite, et de tester par vous-même ces fonctions.

Vous trouverez aussi de l'aide sur Planète-Casio, via les liens suivants :

- www.planete-casio.fr
- <http://www.planet-casio.com/Fr/forums/topic9583-1-%5BTutorial%5D-Les-Str.html>
- <http://www.planet-casio.com/Fr/forums/partie3-1.html>

Bonne continuation !